



Opérateur de dérivation D et macro-commande de dérivation diff

© Pierre Lantagne
Enseignant retraité du Collège de Maisonneuve

La première version de ce document est parue sous la version Maple 6. Ce document est la suite du document « Taux de variation instantané et interprétation graphique ». Ce document présente la distinction entre Opérateur de dérivation D et la macro-commande de dérivation diff . L'opérateur D est un opérateur opérant sur une fonction dont le résultat est une fonction tandis que diff est une macro-commande opérant sur une expression dont le résultat est une expression.

Bonne lecture à tous !

* Ce document Maple est exécutable avec la version 2020.1

Initialisation

```
> restart;  
> with(plots, setoptions):  
setoptions(size=[300,300], axesfont=[times,roman,8]):
```


Question de notation

Définition

f' d'une fonction f peut être définie de la façon suivante, lorsque la limite existe:

$$f'(x) = \lim_{h \rightarrow 0} \frac{f(x+h) - f(x)}{h} \quad (1)$$

Graphiquement, $f'(a)$ correspond à la pente de la tangente à la courbe de f au point $(a, f(a))$



De plus, on dit qu'une fonction f est dérivable sur un intervalle ouvert $]a, b[$ si $f'(x)$ existe $\forall x \in]a, b[$. L'intervalle ouvert pouvant être de la forme $]a, +\infty[$, $]-\infty, b[$, ou $]-\infty, +\infty[$, c'est-à-dire \mathbb{R} .

On dit aussi qu'une fonction f est dérivable sur un intervalle fermé $[a, b]$ si f est dérivable sur l'intervalle ouvert $]a, b[$ et si la dérivée à droite en $x = a$ et la dérivée à gauche en $x = b$ existent, soit que:

$$f'_+(a) = \lim_{h \rightarrow 0^+} \frac{f(a+h) - f(a)}{h} \in \mathbb{R} \text{ et } f'_-(x) = \lim_{h \rightarrow 0^-} \frac{f(b+h) - f(b)}{h} \in \mathbb{R}$$

On peut dire aussi qu'une fonction f est dérivable sur un intervalle semi-ouvert de la forme $[a, b[$, $]a, b]$, $[a, +\infty[$, $]-\infty, b]$.

On se rappellera que $\frac{f(x+h) - f(x)}{h}$ exprime la pente de la sécante à la courbe de f passant par les points $P(x, f(x))$ et $Q(x+h, f(x+h))$ où $h > 0$ est l'accroissement que l'on donne x pour atteindre le point Q de la courbe f . Dans le document « Taux de variation moyen » on a bien constaté qu'en envisageant un accroissement $h < 0$, le calcul de la pente de la sécante passant par les deux mêmes points donnés s'exprime par la même formule.

À l'aide de la formule de la pente d'une droite passant par les points $P(x_1, y_1)$ et $Q(x_2, y_2)$, soit $m = \frac{y_2 - y_1}{x_2 - x_1}$

qu'on résume habituellement par $m = \frac{\Delta y}{\Delta x}$, nous allons obtenir une autre notation pour définition de la dérivée.

Puisque $y_2 - y_1 = f(x+h) - f(x)$ et que $x_2 - x_1 = (x+h) - x = h$, nous avons le même concept formulé de la manière suivante

Définition

La fonction dérivée f' d'une fonction f peut être définie de la façon suivante, lorsque la limite existe:

$$f'(x) = \lim_{\Delta x \rightarrow 0} \frac{f(x + \Delta x) - f(x)}{\Delta x} \quad (2)$$

$$f'(x) = \lim_{\Delta x \rightarrow 0} \frac{\Delta y}{\Delta x}$$

La formulation (1), dite en « h », est appelé notation de Lagrange et la formulation (2), dite en « Δx » est appelé notation de Leibniz. Chaque notation a son utilité dans les développements mathématiques.

Pour différencier l'une de l'autre lorsqu'on parle de la dérivée de la fonction f définie par $y = f(x)$, on réécrit la notation (2) comme suit;

$$\frac{dy}{dx} = \lim_{\Delta x \rightarrow 0} \frac{\Delta y}{\Delta x}$$

Ainsi, pour une fonction f définie par $y = f(x)$ on a

$$y' = f'(x) = \frac{dy}{dx}$$

Pour $y = f(x)$, on utilise aussi la notation suivante $\frac{d}{dx}f(x)$ au lieu de $\frac{df(x)}{dx}$. En isolant $\frac{d}{dy}$ ainsi, on considère cette notation comme un opérateur qui commande la dérivation de $f(x)$.

Opérateur de dérivation D

Nous allons vérifier comment l'évaluateur interprétera formellement le calcul de la limite

$$\lim_{h \rightarrow 0} \frac{f(x+h) - f(x)}{h}$$

pour le cas d'une fonction f quelconque.

```
> unassign('f','x','h');
P:=[x,f(x)];
Q:=[x+h,f(x+h)];
TVM:=(P,Q)->( Q[2]-P[2] ) / ( Q[1]-P[1] );
Limit(TVM(P,Q),h=0)=limit(TVM(P,Q),h=0);
```

$$P := [x, f(x)]$$

$$Q := [x + h, f(x + h)]$$

$$TVM := (P, Q) \mapsto \frac{Q_2 - P_2}{Q_1 - P_1}$$

$$\lim_{h \rightarrow 0} \frac{f(x+h) - f(x)}{h} = D(f)(x) \quad (3.1)$$

Dans le membre de droite de la dernière ligne du bloc précédent, Maple a simplifié $\text{limit}(TVM(P, Q), h=0)$ par l'expression $D(f)(x)$.

D est en un mot réservé en Maple: le nom D désigne l'opérateur fonctionnel de dérivation. Dans ce sens, on écrit $D(f)(x)$

- $D(f)$ pour la fonction dérivée f
- $D(f)(x)$ pour la formule dérivée f(x)

Pour obtenir directement la valeur de la pente de la tangente au tracé de f passant par le point $\left(-\frac{1}{2}, f\left(-\frac{1}{2}\right)\right)$, soit $f\left(-\frac{1}{2}\right)$, il suffit donc de taper la requête $D(f)\left(-\frac{1}{2}\right)$.

$$m_{\text{tan}} = D(f)\left(-\frac{1}{2}\right)$$

Alors-y avec la fonction f utilisée dans le document « Taux de variation instantané », soit la fonction f définie par $f(x) = \frac{1}{x+2}$.

```
> f:=x->1/(x+2);
m[tan]=D(f)(-1/2);
```

$$m_{\text{tan}} = -\frac{4}{9} \quad (3.2)$$

L'équation de cette tangente au tracé de la fonction f passant par le point $\left(-\frac{1}{2}, f\left(-\frac{1}{2}\right)\right)$

$$y = m_{\text{tan}}\left(x - \left(-\frac{1}{2}\right)\right) + f\left(-\frac{1}{2}\right)$$

est alors obtenue par la requête suivante:


```
> Éq_Tangente:=y=D(f)(-1/2)*(x+1/2)+f(-1/2);
```

$$\text{Éq_Tangente} := y = -\frac{4x}{9} + \frac{4}{9} \quad (3.3)$$

Superposons dans un même graphique les tracés de la fonction f et de la tangente au graphique de f passant

par le point $\left(-\frac{1}{2}, f\left(-\frac{1}{2}\right)\right)$. Donnons la couleur navy au tracé de la courbe f et la couleur orange à la tangente.

```
> plot([[x,f(x),x=-4..4],[x,rhs(Éq_Tangente),x=-4..4]],
        discontin=true,
        color=[navy,orange],
        view=[-4..4,-4..4])
```



L'opérateur D maîtrise, bien sûr, les règles de dérivation d'une somme, d'un produit, d'un quotient ainsi que la composition de fonctions.

```
> unassign('f','g');
'D'(f+g) = D(f+g); # Dérivation d'une somme
'D'(f*g) = D(f*g); # Dérivation d'un produit
'D'(f/g) = D(f/g); # Dérivation d'un quotient
'D'(f@g) = D(f@g); # Dérivation en chaîne
```

$$D(f+g) = D(f) + D(g)$$

$$D(fg) = D(f)g + fD(g)$$

$$D\left(\frac{f}{g}\right) = \frac{D(f)}{g} - \frac{fD(g)}{g^2}$$

$$D(f@g) = D(f)@gD(g)$$

(3.4)

Le résultat qu'a donné l'évaluateur quant à la règle de la dérivation du quotient peut vous apparaître un tantinet surprenant. En fait, c'est un résultat tout à fait conforme à la requête qui lui a été acheminée.

En effet, sachez que l'interprète traduit toujours, pour l'évaluateur, la division en une multiplication du dividende par l'inverse multiplicatif du diviseur. Cette traduction est exclusivement syntaxique.

L'évaluateur a donc dérivé en fait la multiplication $f(x)(g^{-1})(x)$

$$D_x\left(\frac{f}{g}\right) = D_x(f \times g^{-1})$$

Bien sûr, le résultat obtenu est équivalent à la formule habituelle de la dérivation du quotient. D'ailleurs, en mode *manu scriptus*, il arrive souvent de transformer un quotient en un produit afin de dériver plus efficacement.

Pour la règle de la dérivation du quotient, Maple la gère sous ce calcul sans souci. Exprimons tout de même que c'est cette règle qu'on utilise en classe,

```
> 'D'(f/g) = simplify(D(f/g));
```

$$D\left(\frac{f}{g}\right) = \frac{D(f)g - fD(g)}{g^2} \quad (3.5)$$

Avant d'introduire la macro-commande `diff`, développons quelques exemples de calculs faits avec l'opérateur `D`.

Exemple 1

Calculer la dérivée $f'(x)$ pour la fonction f définie par $f(x) = (x^4 + 1)^7$.

```
> f:=x-(x^4+1)^7;
'D'(f)(x)=D(f)(x);
```

$$f := x \mapsto (x^4 + 1)^7$$

$$D(f)(x) = 28(x^4 + 1)^6 x^3 \quad (3.1.1)$$

Exemple 2

Calculer la dérivée $g'(x)$ de la fonction g définie par $g(x) = \sqrt{x^2 + \sqrt{3}x}$.

```
> g:=x->sqrt(x^2+sqrt(3*x));
'D'(g)(x)=D(g)(x);
```

$$g := x \mapsto \sqrt{x^2 + \sqrt{3}x}$$

$$D(g)(x) = \frac{2x + \frac{\sqrt{3}}{2}}{2\sqrt{x^2 + \sqrt{3}x}}$$

$$D(g)(x) = \frac{2x + \frac{\sqrt{3}}{2}}{2\sqrt{x^2 + \sqrt{3}x}} \quad (3.2.1)$$

Normalisons ce résultat.

```
> ``=normal(rhs((3.2.1)));
```

$$= \frac{4x^{3/2} + \sqrt{3}}{4\sqrt{x}\sqrt{x^2 + \sqrt{3}x}} \quad (3.2.2)$$

Exemple 3

Calculer $f^{(5)}(x)$, la dérivée successive d'ordre 5 de la fonction f définie par $f(x) = x^7 - 5x^2$.

Pour obtenir cette dérivée successive d'ordre 5, on peut très bien envisager la dérivée de la dérivée de la...

$$f^{(5)}(x) = D(D(D(D(D(f)))))(x)$$

mais il y a plus efficace.

Pour obtenir des dérivées d'ordre supérieur avec l'opérateur de dérivation `D`, il est préférable d'utiliser

l'opérateur de composition fonctionnelle réitérée @@:

$$f^{(5)}(x) = (D@@5)(f)(x)$$

```
> f:=x->x^7-5*x^2;
```

```
'(D@@5)'(f)(x)=(D@@5)(f)(x);
```

$$f := x \mapsto x^7 - 5 \cdot x^2$$

$$D^{(5)}(f)(x) = 2520x^2$$

(3.3.1)

Exemple 4

Calculer la pente de la tangente à la courbe d'équation $y=f(x) = x^2 - \sqrt{x^2 + 2} - 5$ au point $(7, f(7))$.

Il suffit d'évaluer $f(7)$.

```
> f:=x->x^2-sqrt(x^2+2)-5;
```

```
m[tan]=D(f)(7);
```

$$f := x \mapsto x^2 - \sqrt{x^2 + 2} - 5$$

$$m_{\text{tan}} = 14 - \frac{7\sqrt{51}}{51}$$

(3.4.1)

La pente de la tangente à la courbe f au point $(7, f(7))$ est $m_{\text{tan}} = 14 - \frac{7\sqrt{51}}{51}$.

Exemple 5

Déterminer les coordonnées exactes des points sur la courbe d'équation $y=g(x) = 2x^3 + 7x^2 - 28x + 12$ où la tangente à la courbe g est parallèle à l'axe des x .

$g'(x)=0$

```
> g:=x->2*x^3+7*x^2-28*x+12;
```

```
Sol:=solve(D(g)(x)=0);
```

$$g := x \mapsto 2 \cdot x^3 + 7 \cdot x^2 - 28 \cdot x + 12$$

$$\text{Sol} := -\frac{7}{6} + \frac{\sqrt{217}}{6}, -\frac{7}{6} - \frac{\sqrt{217}}{6}$$

(3.5.1)

Les coordonnées exactes des deux points où la tangente à la courbe g est parallèle à l'axe des x sont les suivants.

```
> 'P'[1]=[Sol[1],radnormal(g(Sol[1]))];
```

```
'P'[2]=[Sol[2],radnormal(g(Sol[2]))]
```

$$P_1 = \left[-\frac{7}{6} + \frac{\sqrt{217}}{6}, \frac{2755}{54} - \frac{217\sqrt{217}}{54} \right]$$

$$P_2 = \left[-\frac{7}{6} - \frac{\sqrt{217}}{6}, \frac{2755}{54} + \frac{217\sqrt{217}}{54} \right]$$

(3.5.2)

Exemple 6

Quel est le point sur la courbe d'équation $y=f(x) = 6x^2 + 4x$ pour lequel la pente de la tangente à la courbe est 10 ? Représenter clairement la courbe et la tangente dans un même graphique. Attribuer une couleur différente pour la courbe et pour la tangente.

Obtenons ce point et nommons-le Point.

```
> f:=x->6*x^2+4*x;  
Abscisse:=solve(D(f)(x)=10);  
Point:=[Abscisse,f(Abscisse)];
```

$$f := x \mapsto 6 \cdot x^2 + 4 \cdot x$$

$$\text{Abscisse} := \frac{1}{2}$$

$$\text{Point} := \left[\frac{1}{2}, \frac{7}{2} \right]$$

(3.6.1)

L'équation de la tangente au point $\left(\frac{1}{2}, f\left(\frac{1}{2}\right)\right)$ est de la forme


$$y = m_{\text{tan}}(x - x_0) + y_0$$

$$y = f'\left(\frac{1}{2}\right)\left(x - \frac{1}{2}\right) + f\left(\frac{1}{2}\right)$$

Créons d'abord cette équation qui va vous permettre de superposer les tracés de la courbe f et de la tangente.

```
> Éq:=y=D(f)(Abscisse)*(x-Abscisse)+f(Abscisse);  
plot([[x,f(x),x=-2..2],[x,rhs(Éq),x=-2..2]],color=[navy,orange],  
view=[-3..3,-20..20]);
```

$$\text{Éq} := y = 10x - \frac{3}{2}$$



Macro-commande de dérivation diff

L'opérateur de dérivation D est utilisé pour obtenir la fonction dérivée tandis que la macro-commande `diff`

est utilisée pour dériver une expression. La macro-commande `diff` s'applique donc sur une expression et a comme résultat une expression.

`diff(Expression, variable)`

Soit $y = 3x^2 - 5x + 3$. Calculons y' .

En assignant le trinôme $3x^2 - 5x + 3$ à la variable y , on transpose directement en Maple le sens mathématique que porte chaque variable dans l'équation $y = 3x^2 - 5x + 3$

– la variable x sera considérée comme étant la variable indépendante

– la variable y comme étant la variable dépendante

```
> y:=3*x^2-5*x+3;
```

$$y := 3x^2 - 5x + 3 \quad (4.1)$$

Après cette assignation, la variable y pointe vers le $3x^2 - 5x + 3$ et est donc une variable informatique.

Obtenons maintenant y' à l'aide de la macro-commande `diff`.

```
> diff(y,x);
```

$$6x - 5 \quad (4.2)$$

La macro-commande `diff` est une autre macro-commande qui possède une forme inactive. Cela évite l'usage des apostrophes droit pour la rendre inactive.

L'afficheur pose la dérivée dans la notation de Leibniz.

```
> Diff(y,x);
```

$$\frac{d}{dx} (3x^2 - 5x + 3) \quad (4.3)$$

```
> Diff(y,x)=diff(y,x);
```

$$\frac{d}{dx} (3x^2 - 5x + 3) = 6x - 5 \quad (4.4)$$

Voilà de la lisibilité dans la zone des résultats. La forme inactive d'une macro-commande nous donne la possibilité de bien documenter la zone des résultats.

Il est bien sûr permis d'utiliser les accents graves (voir [?backquotes](#)) pour créer des noms significatifs. Ici aussi, l'usage de ces délimiteurs est plutôt utile pour bien documenter la zone des résultats

```
> y:=(x^4+1)^7+208*x;  
`y'`:=diff(y,x);
```

$$y := (x^4 + 1)^7 + 208x$$
$$y' := 28(x^4 + 1)^6 x^3 + 208 \quad (4.5)$$

y' pointe vers l'expression algébrique $(x^4 + 1)^6 x^3 + 208$ et donc y' n'est en aucun temps le nom d'une fonction. Pour évaluer la dérivée obtenue, il faut utiliser, comme pour toute autre expression, la macro-commande `eval`.

Évaluons donc y' avec $x = 1$.

```
> eval(`y'`,x=1);
```


La macro-commande `eval` possède, elle aussi, une forme inactive qui est intéressante à utiliser pour mieux documenter d'éventuels développements.

```
> Eval('`y`',x=1)=Eval(Diff(y,x),x=1);
``=Eval(diff(y,x),x=1);
``=eval(diff(y,x),x=1);
```

$$\begin{aligned} y' \Big|_{x=1} &= \frac{d}{dx} \left((x^4 + 1)^7 + 208x \right) \Big|_{x=1} \\ &= \left(28(x^4 + 1)^6 x^3 + 208 \right) \Big|_{x=1} \\ &= 2000 \end{aligned} \quad (4.7)$$

Pour obtenir une dérivée successive d'ordre n , il suffit, dans la macro-commande `diff`, de répéter la variable autant de fois que l'ordre de dérivation demandé.

Soit $y = \frac{1}{\sqrt{x}}$. Calculons y''' .

```
> y:=1/sqrt(x);
`y'''`:=diff(y,x,x,x);
```

$$\begin{aligned} y &:= \frac{1}{\sqrt{x}} \\ y''' &:= -\frac{15}{8x^{7/2}} \end{aligned} \quad (4.8)$$

Pour éviter de taper à répétition la variable, il est préférable d'utiliser l'opérateur de séquence `$` (voir [?S](#)). Cet opérateur de séquence sert composer des séquences d'expressions.

```
> $ 5..9;
a[i] $i = 1..6;
x$4;
```

$$\begin{aligned} &5, 6, 7, 8, 9 \\ &a_1, a_2, a_3, a_4, a_5, a_6 \\ &x, x, x, x \end{aligned} \quad (4.9)$$

Calculons la dérivée successive d'ordre 6 de $y = \frac{1}{x+1}$, soit $y^{(6)}$.

```
> y:=x/(x+1);
Diff(y,x$6)=diff(y,x$6);
```

$$\begin{aligned} y &:= \frac{x}{x+1} \\ \frac{d^6}{dx^6} \left(\frac{x}{x+1} \right) &= -\frac{720}{(x+1)^6} + \frac{720x}{(x+1)^7} \end{aligned} \quad (4.10)$$

La dérivée précédente a été effectué par l'évaluateur comme la dérivation du produit $x(x+1)^{-1}$. Rappelons que l'interprète, à l'aide des lois des exposants, traduit toujours pour l'évaluateur, une division par une

multiplication.

Simplifions ce dernier résultat avec la macro-commande `normal`.

`> normal((4.10));`

$$\frac{d^6}{dx^6} \left(\frac{x}{x+1} \right) = -\frac{720}{(x+1)^7} \quad (4.11)$$